

## SubjectJ BNF Grammars

The following are the BNF grammars for the files that SubjectJ will use to perform the compositions. Note that the project is in the *very* pre-alpha stages, and the grammar are not just subject to, but very likely to change.

**The include (.sji) file** This file contains the list of all packages and/or class files that SubjectJ will work with. Any package or class file not listed in this file will not be modified or otherwise touched by SubjectJ. This file is important because even if some of the included classes may not be mentioned in any mappings, SubjectJ will update every class named in this file as necessary so that they correctly interact with any newly composed class(es).

```
⟨Include⟩      ::= ⟨List⟩ ⟨Include⟩
                |   ⟨List⟩

⟨List⟩         ::= package ⟨PathList⟩ ⟨OptionList⟩
                |   class ⟨PathList⟩

⟨PathList⟩     ::= ⟨Path⟩ , ⟨PathList⟩
                |   ⟨Path⟩

⟨Path⟩         ::= ⟨String⟩ . ⟨Path⟩
                |   ⟨String⟩

⟨OptionList⟩   ::= ⟨Option⟩ ⟨OptionList⟩
                |   ⟨Option⟩

⟨Option⟩       ::= exclude ⟨ClassSpecification⟩
                |   recursive

⟨String⟩       ::= c ⟨String⟩
                |   c
```

**The concern (.sjc) file (optional)** This file allows the user to group individual packages or class files into ‘mappings’ to make referring to them easier when defining subjects. In other words, this is the concern mapping file. This file is optional, however; SubjectJ, like Hyper/J, creates a mapping for every included class file (see the includes file above) – the name of the concern will be the Java fully-qualified name. This name can be used in the subjects file (see below) instead if desired, skipping any need for this file.

```
⟨Concerns⟩     ::= ⟨Concern⟩ ⟨Concerns⟩
                |   ⟨Concern⟩

⟨Concern⟩      ::= package ⟨KeyValue⟩
                |   class ⟨KeyValue⟩
                |   interface ⟨KeyValue⟩
                |   method ⟨KeyValue⟩
                |   field ⟨KeyValue⟩

⟨KeyValue⟩     ::= ⟨Path⟩ : ⟨String⟩

⟨Path⟩         ::= ⟨String⟩ . ⟨Path⟩
                |   ⟨String⟩

⟨String⟩       ::= c ⟨String⟩
                |   c
```

**The subject (.sj) file** This is where the action happens; the actual subjects are defined in this file. This grammar is the most likely to change based on requirements and feedback as development continues. As it stands currently, there can be one or more subject files – one for each subjective view of the code. The name on top of the file must be distinct among all subjects – unlike Hyper/J, the names are not specified as commandline arguments; the names from the subject files will be used as the names of the folders in which the composed files will be placed. As noted above, the concerns in this file may be the concerns named in the concerns file (above), or they may simply be Java fully-qualified names. SubjectJ will only perform compositions on the named concerns in this file; all other concerns will be untouched with the exception of being updated to work correctly with the newly composed files.

$\langle \textit{Subject} \rangle ::= \langle \textit{Name} \rangle \langle \textit{Concerns} \rangle \langle \textit{GeneralStrategy} \rangle \langle \textit{SpecificStrategies} \rangle$

$\langle \textit{Name} \rangle ::= \langle \textit{String} \rangle$

$\langle \textit{Concerns} \rangle ::= \langle \textit{Concern} \rangle \langle \textit{Concerns} \rangle$   
 $\quad \quad \quad | \quad \langle \textit{Concern} \rangle$

$\langle \textit{Concern} \rangle ::= \langle \textit{Path} \rangle$   
 $\quad \quad \quad | \quad \langle \textit{String} \rangle$

$\langle \textit{GeneralStrategy} \rangle ::= \text{merge by name}$   
 $\quad \quad \quad | \quad \text{noncorresponding merge}$   
 $\quad \quad \quad | \quad \text{override by name}$

$\langle \textit{SpecificStrategies} \rangle ::= \langle \textit{Strategy} \rangle \langle \textit{SpecificStrategies} \rangle$   
 $\quad \quad \quad | \quad \langle \textit{Strategy} \rangle$

$\langle \textit{Strategy} \rangle ::= \langle \textit{Merge} \rangle$   
 $\quad \quad \quad | \quad \langle \textit{Override} \rangle$   
 $\quad \quad \quad | \quad \dots$

$\langle \textit{Merge} \rangle ::= \text{merge } \langle \textit{Method} \rangle \text{ with } \langle \textit{Method} \rangle$   
 $\quad \quad \quad | \quad \text{merge } \langle \textit{Method} \rangle \text{ with } \langle \textit{Method} \rangle \text{ via } \langle \textit{SummaryMethod} \rangle$   
 $\quad \quad \quad | \quad \text{merge } \langle \textit{Field} \rangle \text{ with } \langle \textit{Field} \rangle \text{ via } \langle \textit{SummaryMethod} \rangle$

$\langle \textit{Override} \rangle ::= \text{override } \langle \textit{Method} \rangle \text{ with } \langle \textit{Method} \rangle$   
 $\quad \quad \quad | \quad \text{override } \langle \textit{Field} \rangle \text{ with } \langle \textit{Field} \rangle$

$\langle \textit{Class} \rangle ::= \langle \textit{Concern} \rangle . \langle \textit{String} \rangle$   
 $\quad \quad \quad | \quad \langle \textit{Concern} \rangle$

$\langle \textit{Method} \rangle ::= \langle \textit{Concern} \rangle . \langle \textit{String} \rangle$   
 $\quad \quad \quad | \quad \langle \textit{Concern} \rangle$

$\langle \textit{Field} \rangle ::= \langle \textit{Concern} \rangle . \langle \textit{String} \rangle$   
 $\quad \quad \quad | \quad \langle \textit{Concern} \rangle$

$\langle \textit{Path} \rangle ::= \langle \textit{String} \rangle . \langle \textit{Path} \rangle$   
 $\quad \quad \quad | \quad \langle \textit{String} \rangle$

$\langle \textit{String} \rangle ::= c \langle \textit{String} \rangle$   
 $\quad \quad \quad | \quad c$